
    OpenAI.Responses

A client library for the OpenAI Responses API with automatic text extraction and cost calculation.
Installation
Add openai_responses to your list of dependencies in mix.exs:
def deps do
  [
    {:openai_responses, "~> 0.6.1"}
  ]
end
Configuration
Set your OpenAI API key using one of these methods:
Environment Variable
export OPENAI_API_KEY="your-api-key"

Application Config
config :openai_responses, :openai_api_key, "your-api-key"
Getting Started
For a comprehensive tutorial and examples, see the interactive tutorial in Livebook.
Advanced Examples
Simple terminal chat
defmodule Chat do
  alias OpenAI.Responses

  def run do
    IO.puts("Simple AI Chat (type /exit or /quit to end)")
    IO.puts("=" |> String.duplicate(40))

    loop(nil)
  end

  defp loop(previous_response) do
    input = IO.gets("\nYou: ") |> String.trim()

    case input do
      cmd when cmd in ["/exit", "/quit"] ->
        IO.puts("\nGoodbye!")

      _ ->
        IO.write("\nAI: ")

        # Use previous response for context, or create new conversation
        response = if previous_response do
          # Continue conversation with context
          Responses.create!(
            previous_response,
            input: input,
            stream: Responses.Stream.delta(&IO.write/1)
          )
        else
            # First message - start new conversation
            Responses.create!(
              input: input,
              stream: Responses.Stream.delta(&IO.write/1)
            )
        end

        IO.puts("")  # Add newline after response
        loop(response)
    end
  end
end

# Run the chat
Chat.run()
Streaming with Structured Output
# Stream a JSON response with structured output
Responses.stream(
  input: "List 3 programming languages with their year of creation",
  model: "gpt-4o-mini",
  schema: %{
    languages: {:array, %{
      name: :string,
      year: :integer,
      paradigm: {:string, description: "Main programming paradigm"}
    }}
  }
)
|> Responses.Stream.json_events()
|> Enum.each(&IO.puts/1)
Array Schemas (New in 0.6.0)
Arrays can now be used directly at the root level of schema definitions:
# Define an array schema at the root level
{:ok, response} = Responses.create(
  input: "List 3 interesting facts about space exploration",
  schema: {:array, %{
    fact: :string,
    year: {:integer, description: "Year of the event"},
    significance: {:string, description: "Why this fact is important"}
  }}
)

# The response.parsed will be an array directly:
# [
#   %{"fact" => "First satellite launch", "year" => 1957, "significance" => "Started the space age"},
#   %{"fact" => "Moon landing", "year" => 1969, "significance" => "First humans on another celestial body"},
#   %{"fact" => "ISS construction", "year" => 1998, "significance" => "Permanent human presence in space"}
# ]
The library automatically handles OpenAI's requirement that the root level must be an object by wrapping and unwrapping arrays transparently.
Cost Tracking with High Precision
{:ok, response} = Responses.create("Explain quantum computing")

# All cost values are Decimal for precision
IO.inspect(response.cost)
# => %{
#      input_cost: #Decimal<0.0004>,
#      output_cost: #Decimal<0.0008>,
#      total_cost: #Decimal<0.0012>,
#      cached_discount: #Decimal<0>
#    }

# Convert to float if needed
total_in_cents = response.cost.total_cost |> Decimal.mult(100) |> Decimal.to_float()
Documentation
	API Documentation
	Interactive Tutorial
	GitHub Repository



  

    Changelog

The format is based on Keep a Changelog,
and this project adheres to Semantic Versioning.
0.7.0
Added
	LLM Options Preservation in create/2 - When creating follow-up responses using create/2 with a Response object, the following LLM options are now automatically preserved from the previous response unless explicitly overridden:
	model - The model used for generation
	text - Text generation settings including verbosity levels
	reasoning - Reasoning settings including effort levels (e.g., "low", "medium", "high")

This makes it easier to maintain consistent generation settings across conversation chains:
# Initial response with specific settings
first = Responses.create!(
  input: "Complex question",
  model: "gpt-5-mini",
  reasoning: %{effort: "high"},
  text: %{verbosity: "low"}
)

# Follow-up inherits all LLM settings
followup = Responses.create!(first, input: "Tell me more")
# Uses gpt-5-mini, high reasoning, and low verbosity

# Partial override - only change reasoning
another = Responses.create!(first, 
  input: "Another question",
  reasoning: %{effort: "low"}
)
# Still uses gpt-5-mini and low verbosity, but with low reasoning


Changed
	Refactored model preservation logic - The model preservation functionality is now part of a more general LLM options preservation system, making the code more maintainable and extensible

0.6.1
Added
	GPT-5 model pricing support - Added pricing information for the newly released GPT-5 model family:	GPT-5: $1.25/$0.125/$10.00 per million tokens (input/cached/output)
	GPT-5-mini: $0.25/$0.025/$2.00 per million tokens
	GPT-5-nano: $0.05/$0.005/$0.40 per million tokens
	Includes support for all model variants (base models, dated versions, and chat-latest)



0.6.0
Added
	Automatic array wrapping for Structured Outputs - Arrays can now be used at the root level of schema definitions. The library automatically wraps arrays in objects to comply with OpenAI's requirement that the root level must be an object, and unwraps them in the response:Responses.create!(
  input: "list facts about 3 US presidents", 
  schema: {:array, %{
    name: :string,
    birth_year: :integer,
    little_known_facts: {:array, {:string, max_items: 2}}
  }}
)
# Returns an array directly, not wrapped in an object
	Schema.build_output/1 detects array schemas and wraps them in an object with an "items" property
	Response.extract_json/1 detects wrapped arrays and automatically unwraps them
	Objects that naturally have an "items" property are not affected



0.5.3
Fixed
	Duplicate assistant response handling - Fixed an issue where the OpenAI API sometimes returns multiple nearly identical assistant responses by only taking the first assistant response in Response.extract_text/1

0.5.2
Changed
	Internal refactoring - Improved input handling to consistently accept both maps and keyword lists with atom or string keys across all functions
	Documentation - Added comprehensive documentation for the :schema option in create/1 function, explaining how to use structured output with examples

Fixed
	Fixed Dialyzer issues
	Fixed Credo warnings for code quality

0.5.1
Added
	Map support for options - Multiple functions now accept maps in addition to keyword lists for options, providing more flexibility and consistency with Elixir conventions:	create/1 and create/2 - Create responses with map or keyword list options
	stream/1 - Stream responses using map or keyword list options
	run/2 - Run function calling conversations with map or keyword list options
	OpenAI.Responses.Stream.stream/1 and OpenAI.Responses.Stream.stream_with_callback/2 - Stream functions also accept maps
	Maps with mixed atom and string keys are supported
	Example: Responses.create(%{input: "Hello", model: "gpt-4o"})



0.5.0
Added
	Union type support via anyOf - Schema.build_output/1 and Schema.build_function/3 now support union types through the anyOf specification. This allows defining properties that can be one of multiple types:	{:anyOf, [:string, :number]} - tuple syntax for simple unions
	[:anyOf, [:string, :number]] - list syntax for simple unions
	Complex unions with objects and nested structures are fully supported


	LLM usage guide - Added comprehensive usage-rules.md documentation with detailed examples and best practices for using the OpenAI.Responses library with LLM agents
	Manual function calling with call_functions/2 - Added public function to execute function calls from a response and format results for the API. This enables custom workflows where users need to intercept, modify, or add context to function results before continuing the conversation. Function return values are passed through without string conversion, so they must be JSON-encodable (maps, lists, strings, numbers, booleans, nil)
	Error module with retry support - Added OpenAI.Responses.Error exception module that:	Represents OpenAI API errors with fields for message, code, param, type, and HTTP status
	Provides retryable?/1 function to determine if errors can be retried
	Supports both HTTP status codes (429, 500, 503) and Req transport errors (timeout, closed)



Fixed
	Model preservation in follow-up responses - create/2 now correctly preserves the model from the previous response when creating follow-ups, instead of defaulting to the standard model. The model can still be explicitly overridden when needed

Changed
	Package metadata - Updated mix.exs to include additional files in the hex package distribution

0.4.2
Added
	List format support for Schema definitions - Schema.build_output/1 and Schema.build_function/3 now accept list syntax [type, options] in addition to tuple syntax {type, options}. This provides more flexibility when defining schemas:	[:string, %{description: "Full name"}] - list with map options
	[:string, [description: "Full name"]] - list with keyword list options
	[:array, :string] - list format for arrays
	All formats can be mixed within the same schema definition


	Pricing for o3 pro model

Changed
	Cost calculation for unknown models - When pricing data is not available for a model, Response.calculate_cost/1 now returns zero costs for all categories instead of nil. This provides a consistent cost structure regardless of whether the model has pricing information.
	Schema internals refactored - The schema building logic now uses a two-phase approach (normalize then build) making it more maintainable and extensible. This is an internal change that doesn't affect the public API.
	Pricing for o3 model updated to reflect 80% price decrease

0.4.1
Added
	Support for string tuple syntax in Schema arrays - Schema.build_property/1 now accepts {"array", item_type} in addition to {:array, item_type} for defining array types in schemas
	Support for string keys in Schema definitions - Schema.build_output/1 and Schema.build_function/3 now accept lists with string keys in addition to keyword lists with atom keys. This enables schemas to be stored in databases or other persistence layers where atoms are converted to strings, and then reused without conversion back to atoms

0.4.0
Major refactoring to make the codebase more modular and maintainable, and simplify the API.
Breaking Changes
	Removed :client option from create/1
	Removed Responses.Helpers module - cost is now calculated automatically for models with known pricing
	Removed image helpers for now until a better API can be designed
	Removed delete/2, get/2, and list_input_items/2 - use the low-level request/1 function instead
	Removed parse/2 - use create/1 with a :schema option instead
	Removed collect_stream/1
	Changed schema syntax - replaced Schema.object/2, Schema.string/1 etc. with simple maps

Added
	create/2 can be used to follow up on a previous response
	create!/1 and create!/2 raise an error if the response is not successful
	run/2 and run!/2 functions for automatic function calling loops	Takes a list of available functions and automatically handles function calls
	Continues conversation until a final response without function calls is received
	Returns a list of all responses generated during the conversation
	Supports both map and keyword list function definitions


	list_models/0 and list_models/1 to return the list of available OpenAI models
	:schema option to create/1 to specify the schema of the response body
	Schema.build_output/1 to build the output schema for the response body
	Schema.build_function/3 to build the function schema for tool calling
	:stream callback function option to create/1 that will be called on every event
	json_events/1 helper to Responses.Stream for streaming JSON events with Jaxon

Changed
	create/1 now returns a Response struct containing the generated response body, text, parsed JSON, and cost information
	text_deltas/1 is moved to Responses.Stream

0.3.0
Breaking Changes
	Removed positional parameters from create, stream, and parse functions. Use keyword lists or maps for parameters instead.

Instead of OpenAIResponses.create("model", "prompt"), use OpenAIResponses.create(model: "model", prompt: "prompt").
	Changed parse response format. The structure of parse return has changed: instead of the parsed response,
it now returns a map with the following keys:
	parsed: the parsed result
	raw_response: the raw response received from OpenAI, which can be used to e.g. calculate the cost
	token_usage: information about the token usage


	Removed Config module. Configuration handling has been streamlined. (Commit: 88b812e)

	Removed parse_stream function. This function was not working as intended, and is removed in this release. (Commit: 9723ccb)

	Removed explicit instructions message in parse. This should now be handled by the developer. (Commit: f8addc0)


Added
	Cost calculations for various models including 4.1 models and chatgpt-4o-latest. (Commits: 2985d94, 70d68ad, 4f2cdc3)
	Option to log requests for debugging or monitoring purposes. (Commit: d4d56b5)

Changed
	Increased default timeout to 60 seconds from a previous value. (Commit: 4a01b66)

Fixed
	Fixed tests across multiple commits to ensure functionality and improve coverage. (Commits: c78858c, 18c535a, 804522e, c594688, d637a73)

Documentation
	Updated documentation to reflect new features, response usage, and tutorial content. (Commits: d1bc204, f07d990, 03031e6, a4ac509)



  

    Tutorial

Mix.install([
  {:openai_responses, path: "~/src/responses"},
  {:kino, "~> 0.11.0"}
])
Introduction
OpenAI.Responses is a Small Development Kit for the OpenAI Responses API. It can get you started in no time, and automatically supports conversation state, structured responses, streaming, function calls, and cost calculations.
If you want an industrial-grade library that supports multiple providers, you should instead consider one of LangChain, OpenaiEx, or Instructor. If, however, you want to start as quickly as possible without learning a new framework first, and are ready to trade the flexibility of choosing an LLM provider for the conveniences of OpenAI's latest API, you've come to the right place!
Basic usage
The basic usage is simple: set the OPENAI_API_KEY environment variable and add openai_responses to your mix.exs
def deps do
  [
    # ...
    {:openai_responses, "~> 0.5.0"}
  ]
end
is enough to get you started:
alias OpenAI.Responses

{:ok, response} = Responses.create("Write me a haiku about Elixir")
# -> {:ok, %OpenAI.Responses.Response{text: ..., ...}}

IO.puts(response.text)
create(s) when is_binary(s) is a shortcut for create(input: s). In general, create/1 takes a keyword list with anything that Create a model response supports. create!/1 is a version of create/1 that raises on errors:
response =
  Responses.create!(
    input: [
      %{role: :developer, content: "Talk like a pirate."},
      %{role: :user, content: "Write me a haiku about Elixir"}
    ],
    model: "gpt-5-mini",
    reasoning: %{effort: :low}
  )

IO.puts("#{response.text}\n\nCost: $#{response.cost.total_cost}")
There is also create/2 and create!/2 that take %Responses.Response{} as a first argument and keyword list as a second to automatically handle the conversation state:
Responses.create!(
  input: [
    %{role: :developer, content: "Talk like a pirate."},
    %{role: :user, content: "Write me a haiku about Elixir"}
  ]
)
|> Responses.create!(input: "Which programming language is this haiku about?")
|> Map.get(:text)
|> IO.puts()
Notice that the follow-up response also talks like a pirate, as it inherits the model, developer settings, etc. from the initial response.
Structured Output
One of the coolest features of the latest generation of OpenAI models is their ability to output Structured Output that precisely matches the supplied JSON schema. create/1 accepts a schema: option that allows easy creation of such schema in the required format.
The return result will automatically parse the response into %Response{parsed: object}.
alias OpenAI.Responses

Responses.create!(
  input: "List facts about first 3 U.S. Presidents",
  schema: %{
    presidents:
      {:array,
       %{
         name: :string,
         birth_year: :integer,
         little_known_facts: {:array, {:string, max_items: 2}}
       }}
  }
)
|> Map.get(:parsed)
|> Map.get("presidents")
The above example should give you enough to get started, and you can read the documentation about supported schemas.
Streaming
Streaming model responses is a great way to increase interactivity of your application. There are two ways to support streaming with OpenAI.Responses:
	By adding a callback: callback_fn option to create/1. Here callback_fn/1 takes a map %{event: type, data: data} and should return either :ok to continue streaming or {:error, reason} to stop.
The call will be blocked until the streaming ends but will otherwise return the same
%Response{} structure.

	By calling stream/1, which returns an Elixir Stream of %{event: type, data: data} objects.


For the supported event types and data format, refer to the Streaming Responses API docs.
Here is an example of how this works. We use the Responses.Stream.text_deltas helper to transform the stream of events into a stream of text chunks, and use Kino.Frame to demonstrate interactive output updates in a Livebook.
alias OpenAI.Responses

frame = Kino.Frame.new()
Kino.render(frame)

Responses.stream(
  input: """
  Write a short fairy tale about an Elixir developer who tried to use Java,
  and about the horrors that have ensued.
  """,
  temperature: 0.7
)
|> Responses.Stream.text_deltas()
|> Stream.each(fn delta ->
  Kino.Frame.append(frame, Kino.Markdown.new(delta, chunk: true))
end)
|> Stream.run()

:done
There is also a Responses.Stream.json_events/1 helper, which uses the Jaxon library to stream JSON events:
Responses.stream(
  input: "Tell me about first 2 U.S. Presidents",
  schema: %{presidents: {:array, %{name: :string, birth_year: :integer}}}
)
|> Responses.Stream.json_events()
|> Stream.each(&IO.inspect/1)
|> Stream.run()
Tools
Using OpenAI built-in tools, for example Web Search, is simple: just add a tools: parameter to create/1:
alias OpenAI.Responses

Responses.create!(
  input: "Summarize in 3 paragraphs a positive news story from today",
  tools: [%{type: "web_search_preview"}]
)
|> Map.get(:text)
|> IO.puts
For Function calling, you can provide a function description. The Responses.Schema.build_function/3 helper makes this easier:
# Using the build_function helper
weather_tool = Responses.Schema.build_function(
  "get_weather",
  "Get current temperature for a given location",
  %{location: {:string, description: "City and country e.g. Bogotá, Colombia"}}
)

response = Responses.create!(
  input: "What is the weather like in Paris today?",
  tools: [weather_tool]
)

response.function_calls
The resulting %Response{} has a :function_calls field populated with the function calls requested by the model. Our app can now call each function, and we can provide results back to the model:
response
|> Responses.create!(
  input: [%{
    type: "function_call_output",
    call_id: response.function_calls |> List.first() |> Map.get(:call_id),
    output: "15C"
  }]
)
|> Map.get(:text)
Automating Function Calls with run/2
The Responses.run/2 function automates the process of handling function calls. It will repeatedly call your functions and feed the results back to the model until a final response is achieved:
# Define available functions
functions = %{
  "get_weather" => fn %{"location" => location} ->
    # In a real app, this would call a weather API
    case location do
      "Paris" -> "15°C, partly cloudy"
      "London" -> "12°C, rainy"
      "New York" -> "8°C, sunny"
      _ -> "Weather data not available"
    end
  end,
  "get_time" => fn %{"timezone" => timezone} ->
    # In a real app, this would get actual time for timezone
    case timezone do
      "Europe/Paris" -> "14:30"
      "Europe/London" -> "13:30"
      "America/New_York" -> "08:30"
      _ -> "Unknown timezone"
    end
  end
}

# Define function tools
weather_tool = Responses.Schema.build_function(
  "get_weather",
  "Get current weather for a location",
  %{location: {:string, description: "City name"}}
)

time_tool = Responses.Schema.build_function(
  "get_time",
  "Get current time in a timezone",
  %{timezone: {:string, description: "Timezone like Europe/Paris"}}
)

# Run the conversation with automatic function calling
responses = Responses.run(
  [
    input: "What's the weather and time in Paris?",
    tools: [weather_tool, time_tool]
  ],
  functions
)

# The last response contains the final answer
responses |> List.last() |> Map.get(:text) |> IO.puts()

# You can also inspect all intermediate responses
IO.puts("\nTotal responses: #{length(responses)}")
The run/2 function returns a list of all responses generated during the conversation. This allows you to:
	Track the conversation flow
	See what functions were called
	Calculate total costs across all API calls
	Debug issues in function calling

There's also a run!/2 variant that raises on errors instead of returning error tuples.

Client for OpenAI Responses API.
This module provides a simple interface for creating AI responses with support for:
	Text and structured output generation
	Streaming responses with Server-Sent Events (SSE)
	Automatic cost calculation for all API calls
	JSON Schema-based structured outputs

Available Functions
	create/1 and create/2 - Create AI responses (synchronous or streaming)
	create!/1 and create!/2 - Same as create but raises on error
	run/2 and run!/2 - Run conversations with automatic function calling
	call_functions/2 - Execute function calls and format results for the API
	stream/1 - Stream responses as an Enumerable
	list_models/0 and list_models/1 - List available OpenAI models
	request/1 - Low-level API request function

Configuration
Set your OpenAI API key via:
	Environment variable: OPENAI_API_KEY
	Application config: config :openai_responses, :openai_api_key, "your-key"

Examples
See the tutorial for comprehensive examples and usage patterns.

        call_functions(function_calls, functions)

      


        Execute function calls from a response and format the results for the API.



    


    
      
        create(options)

      


        Create a new response.



    


    
      
        create(previous_response, options)

      


        Create a response based on a previous response.



    


    
      
        create!(options)

      


        Same as create/1 but raises an error on failure.



    


    
      
        create!(previous_response, options)

      


        Same as create/2 but raises an error on failure.



    


    
      
        list_models(match \\ "")

      


        List available models.



    


    
      
        request(options)

      


        Request a response from the OpenAI API.



    


    
      
        run(options, functions)

      


        Run a conversation with automatic function calling.



    


    
      
        run!(options, functions)

      


        Same as run/2 but raises an error on failure.



    


    
      
        stream(options)

      


        Stream a response from the OpenAI API as an Enumerable.



    





      


      
        Functions

        


  
    
      
    
    
      call_functions(function_calls, functions)



        
          
        

    

  


  

Execute function calls from a response and format the results for the API.
Takes the function_calls from a response and a map/keyword list of functions,
executes each function with its arguments, and returns the formatted results
ready to be used as input for the next API call.
Parameters
	function_calls - The function_calls array from a Response struct
	functions - A map or keyword list where:	Keys are function names (as atoms or strings)
	Values are functions that accept the parsed arguments and return the result



Returns
Returns a list of formatted function outputs suitable for use as input to create/2.
Important: Function return values must be JSON-encodable. This means they should
only contain basic types (strings, numbers, booleans, nil), lists, and maps. Tuples,
atoms (except true, false, and nil), and other Elixir-specific types are not
supported by default unless they implement the Jason.Encoder protocol.
Examples
# Get a response with function calls
{:ok, response} = Responses.create(
  input: "What's the weather in Paris and what time is it?",
  tools: [weather_tool, time_tool]
)

# Define the actual function implementations
functions = %{
  "get_weather" => fn %{"location" => location} ->
    # Returns a map (JSON-encodable)
    %{temperature: 22, unit: "C", location: location}
  end,
  "get_time" => fn %{} ->
    # Returns a string (JSON-encodable)
    DateTime.utc_now() |> to_string()
  end
}

# Execute the functions and get formatted output
outputs = Responses.call_functions(response.function_calls, functions)

# Continue the conversation with the function results
{:ok, final_response} = Responses.create(response, input: outputs)
Error Handling
If a function is not found or raises an error, the output will contain
an error message instead of the function result.

Create a new response.
When the argument is a string, it is used as the input text.
Otherwise, the argument is expected to be a keyword list or map of options that OpenAI expects,
such as input, model, temperature, max_tokens, etc.
LLM Options Preservation with previous_response_id
The OpenAI API always requires a model parameter, even when using previous_response_id.
When using create/1 with manual previous_response_id:
	If no model is specified, the default model is used
	LLM options (model, text, reasoning) from the previous response are NOT automatically inherited

When using create/2 with a Response object:
	LLM options (model, text, reasoning) from the previous response ARE automatically inherited

	You can override any of them by explicitly specifying different values
  # Manual previous_response_id - uses defaults if not specified
  Responses.create(input: "Hello", previous_response_id: "resp_123")
  # Manual previous_response_id - with explicit options
  Responses.create(input: "Hello", previous_response_id: "resp_123", model: "gpt-4.1")
  # Using create/2 - automatically inherits LLM options from previous response
  Responses.create(previous_response, input: "Hello")
  # Using create/2 - with reasoning effort preserved (requires model that supports reasoning)
  first = Responses.create!(input: "Question", model: "gpt-5-mini", reasoning: %{effort: "high"})
  followup = Responses.create!(first, input: "Follow-up")  # Inherits gpt-5-mini and high reasoning


Examples
# Using a keyword list
Responses.create(input: "Hello", model: "gpt-4.1", temperature: 0.7)

# Using a map
Responses.create(%{input: "Hello", model: "gpt-4.1", temperature: 0.7})

# String shorthand
Responses.create("Hello")
Structured Output with :schema
Pass a schema: option to get structured JSON output from the model.
The schema is defined using a simple Elixir syntax that is converted to JSON Schema format.
Both maps and keyword lists with atom or string keys are accepted for all options:
# Using a map with atom keys
Responses.create(%{
  input: "Extract user info from: John Doe, username @johndoe, john@example.com",
  schema: %{
    name: :string,
    username: {:string, pattern: "^@[a-zA-Z0-9_]+$"},
    email: {:string, format: "email"}
  }
})

# Using a keyword list
Responses.create(
  input: "Extract product details",
  schema: [
    product_name: :string,
    price: :number,
    in_stock: :boolean,
    tags: {:array, :string}
  ]
)

# Arrays at the root level (new in 0.6.0)
Responses.create(
  input: "List 3 US presidents with facts",
  schema: {:array, %{
    name: :string,
    birth_year: :integer,
    achievements: {:array, :string}
  }}
)
# Returns an array directly in response.parsed

# Mixed keys (atoms and strings) are supported
Responses.create(%{
  "input" => "Analyze this data",
  :schema => %{
    "result" => :string,
    :confidence => :number
  }
})
The response will include a parsed field with the extracted structured data.
See OpenAI.Responses.Schema for the full schema syntax documentation.
Streaming
Pass a stream: option with a callback function to stream the response.
The callback receives results wrapped in {:ok, chunk} or {:error, reason} tuples:
Responses.create(
  input: "Write a story",
  stream: fn
    {:ok, %{event: "response.output_text.delta", data: %{"delta" => text}}} ->
      IO.write(text)
      :ok
    {:error, reason} ->
      IO.puts("Stream error: #{inspect(reason)}")
      :ok  # Continue despite errors
    _ ->
      :ok
  end
)
The callback should return :ok to continue or {:error, reason} to stop the stream.
For simpler text streaming, use the delta/1 helper:
Responses.create(
  input: "Write a story",
  stream: Responses.Stream.delta(&IO.write/1)
)
If no model is specified, the default model is used.

Create a response based on a previous response.
This allows creating follow-up responses that maintain context from a previous response.
The previous response's ID is automatically included in the request.
Options can be provided as either a keyword list or a map.
Preserved Options
The following options are automatically preserved from the previous response unless explicitly overridden:
	model - The model used for generation
	text - Text generation settings (including verbosity)
	reasoning - Reasoning settings (including effort level)

Examples
{:ok, first} = Responses.create("What is Elixir?")

# Using keyword list
{:ok, followup} = Responses.create(first, input: "Tell me more about its concurrency model")

# Using map
{:ok, followup} = Responses.create(first, %{input: "Tell me more about its concurrency model"})

# With reasoning effort preserved (requires model that supports reasoning)
{:ok, first} = Responses.create(input: "Complex question", model: "gpt-5-mini", reasoning: %{effort: "high"})
{:ok, followup} = Responses.create(first, input: "Follow-up")  # Inherits gpt-5-mini and high reasoning effort

Same as create/1 but raises an error on failure.
Returns the response directly instead of an {:ok, response} tuple.
Examples
response = Responses.create!("Hello, world!")
IO.puts(response.text)

Same as create/2 but raises an error on failure.
Returns the response directly instead of an {:ok, response} tuple.
Examples
first = Responses.create!("What is Elixir?")
followup = Responses.create!(first, input: "Tell me more")

  



    

  
    
      
    
    
      list_models(match \\ "")



        
          
        

    

  


  

List available models.
Accepts an optional match string to filter by model ID.

Request a response from the OpenAI API.
Used as a building block by other functions in this module.
Accepts that same arguments as Req.request/1.
You should provide url, json, method, and other options as needed.

Run a conversation with automatic function calling.
This function automates the process of handling function calls by repeatedly calling the
provided functions and feeding their results back to the model until a final response
without function calls is received.
Parameters
	options - Keyword list or map of options to pass to create/1
	functions - A map or keyword list where:	Keys are function names (as atoms or strings)
	Values are functions that accept the parsed arguments and return the result



Returns
Returns a list of all responses generated during the conversation, in chronological order.
The last response in the list will be the final answer without function calls.
Examples
# Define available functions
functions = %{
  "get_weather" => fn %{"location" => location} ->
    # Simulate weather API call
    "The weather in #{location} is 72°F and sunny"
  end,
  "get_time" => fn %{} ->
    DateTime.utc_now() |> to_string()
  end
}

# Create function tools
weather_tool = Responses.Schema.build_function(
  "get_weather",
  "Get current weather for a location",
  %{location: :string}
)

time_tool = Responses.Schema.build_function(
  "get_time",
  "Get the current UTC time",
  %{}
)

# Run the conversation (with keyword list)
responses = Responses.run(
  [input: "What's the weather in Paris and what time is it?",
   tools: [weather_tool, time_tool]],
  functions
)

# Or with map
responses = Responses.run(
  %{input: "What's the weather in Paris and what time is it?",
    tools: [weather_tool, time_tool]},
  functions
)

# The last response contains the final answer
final_response = List.last(responses)
IO.puts(final_response.text)

Same as run/2 but raises an error on failure.
Returns the list of responses directly instead of an {:ok, responses} tuple.

Stream a response from the OpenAI API as an Enumerable.
Returns a Stream that yields chunks with event and data keys.
Options can be provided as either a keyword list or a map.
Examples
# Stream and handle all results
for result <- Responses.stream("Tell me a story") do
  case result do
    {:ok, chunk} -> IO.inspect(chunk)
    {:error, reason} -> IO.puts("Error: #{inspect(reason)}")
  end
end

# Process only text deltas, ignoring errors
Responses.stream("Write a poem")
|> Stream.filter(fn
  {:ok, %{event: "response.output_text.delta"}} -> true
  _ -> false
end)
|> Stream.map(fn {:ok, chunk} -> chunk.data["delta"] end)
|> Enum.each(&IO.write/1)

# Accumulate all text with error handling (using map)
result = Responses.stream(%{input: "Explain quantum physics"})
         |> Enum.reduce(%{text: "", errors: []}, fn
           {:ok, %{event: "response.output_text.delta", data: %{"delta" => delta}}}, acc ->
             %{acc | text: acc.text <> delta}
           {:error, reason}, acc ->
             %{acc | errors: [reason | acc.errors]}
           _, acc ->
             acc
         end)

  


        

      


  

    
OpenAI.Responses.Pricing 
    



      
Pricing information for OpenAI models.
Prices are in USD per million tokens, stored as Decimal values for precision.
Source: https://platform.openai.com/docs/pricing, last updated on 2025-06-03.

      


      
        Summary


  
    Functions
  


    
      
        get_pricing(model)

      


        Get pricing information for a specific model.



    


    
      
        list_models()

      


        List all available models with pricing.

Get pricing information for a specific model.
Returns a map with :input, :cached_input, and :output prices per million tokens as Decimal values,
or nil if the model is not found.

List all available models with pricing.

  


        

      


  

    
OpenAI.Responses.Response 
    



      
Represent a response from the OpenAI API.
The Response struct contains the following fields:
	text: The extracted text from the response body.
	parsed: The parsed response body.
	parse_error: A map containing error messages if parsing failed.
	function_calls: An array of extracted function calls from the response.
	body: The raw response body.
	cost: The calculated cost of the response in USD.

All of the functions in this module act like "plugs", meaning they take a response as input and return a modified response as output.
The extract_text/1 function extracts the text from the response body, updating the text field of the response.
The extract_json/1 function extracts the data from the response body if it is a structured response.
The extract_function_calls/1 function extracts function calls from the response body.
The calculate_cost/1 function calculates the cost of the response based on token usage and model pricing.

      


      
        Summary


  
    Functions
  


    
      
        calculate_cost(response)

      


        Calculate the cost of the response based on token usage and model pricing.



    


    
      
        extract_function_calls(response)

      


        Extract function calls from the response body.



    


    
      
        extract_json(response)

      


        Extract the data from the response body if it is a structured response.



    


    
      
        extract_text(response)

      


        Extract the text from the response body.

Calculate the cost of the response based on token usage and model pricing.
Returns the response with the cost field updated with a map containing:
	input_cost: Cost for input tokens in USD (as Decimal)
	output_cost: Cost for output tokens in USD (as Decimal)
	total_cost: Total cost in USD (as Decimal)
	cached_discount: Amount saved from cached tokens in USD (as Decimal) (if applicable)

If pricing information is not available for the model, or usage information
is missing, the cost field will be set to zero for all cost categories.

Extract function calls from the response body.
Function calls are extracted from the response.body["output"] array and transformed
into a more convenient format with parsed arguments.

  



  
    
      
    
    
      extract_json(response)



        
          
        

    

  


  

Extract the data from the response body if it is a structured response.
Automatically extracts the text from the response body if it is not already extracted.
If the schema was originally an array at the root level, it will be automatically
unwrapped from the temporary object wrapper.

  



  
    
      
    
    
      extract_text(response)



        
          
        

    

  


  

Extract the text from the response body.
Only extracts text from the first assistant response to handle cases where 
the API returns duplicate assistant responses.

  


        

      


  

    
OpenAI.Responses.Schema 
    



      
Helper module for defining structured output schemas and function calling tools.
Converts simple Elixir syntax into JSON Schema format for structured outputs and function parameters.
Array Support (New in 0.6.0)
Arrays can now be used at the root level of schema definitions. The library automatically
handles OpenAI's requirement that the root level must be an object by wrapping arrays
in a temporary object structure and unwrapping them in the response.
Examples
Structured Output Schema with Object
iex> Responses.Schema.build_output(%{
...>   name: {:string, description: "The name of the user"},
...>   username: {:string, description: "The username of the user. Must start with @", pattern: "^@[a-zA-Z0-9_]+$"},
...>   email: {:string, description: "The email of the user", format: "email"}
...> })
%{
  "name" => "data",
  "type" => "json_schema",
  "strict" => true,
  "schema" => %{
    "type" => "object",
    "properties" => %{
      "name" => %{
        "type" => "string",
        "description" => "The name of the user"
      },
      "username" => %{
        "type" => "string",
        "description" => "The username of the user. Must start with @",
        "pattern" => "^@[a-zA-Z0-9_]+$"
      },
      "email" => %{
        "type" => "string",
        "description" => "The email of the user",
        "format" => "email"
      }
    },
    "additionalProperties" => false,
    "required" => ["name", "username", "email"]
  }
}
Structured Output Schema with Array at Root
iex> Responses.Schema.build_output({:array, %{
...>   title: :string,
...>   completed: :boolean,
...>   priority: {:integer, minimum: 1, maximum: 5}
...> }})
%{
  "name" => "data",
  "type" => "json_schema",
  "strict" => true,
  "schema" => %{
    "type" => "object",
    "properties" => %{
      "items" => %{
        "type" => "array",
        "items" => %{
          "type" => "object",
          "properties" => %{
            "title" => %{"type" => "string"},
            "completed" => %{"type" => "boolean"},
            "priority" => %{"type" => "integer", "minimum" => 1, "maximum" => 5}
          },
          "additionalProperties" => false,
          "required" => ["completed", "priority", "title"]
        }
      }
    },
    "additionalProperties" => false,
    "required" => ["items"]
  }
}
When using array schemas, the response will be automatically unwrapped so that
response.parsed contains the array directly, not wrapped in an object.
Function Calling Tool
iex> Responses.Schema.build_function("get_weather", "Get current temperature for a given location.", %{
...>   location: {:string, description: "City and country e.g. Bogotá, Colombia"}
...> })
%{
  "name" => "get_weather",
  "type" => "function",
  "strict" => true,
  "description" => "Get current temperature for a given location.",
  "parameters" => %{
    "type" => "object",
    "properties" => %{
      "location" => %{
        "type" => "string",
        "description" => "City and country e.g. Bogotá, Colombia"
      }
    },
    "additionalProperties" => false,
    "required" => ["location"]
  }
}

      


      
        Summary


  
    Functions
  


    
      
        build_function(name, description, parameters)

      


        Builds a function calling tool schema.



    


    
      
        build_output(fields)

      


        Builds a structured output schema from a simple Elixir map or keyword list format.

Builds a function calling tool schema.
Parameters
	name - The function name
	description - A description of what the function does
	parameters - A map or keyword list of parameter definitions (same format as build_output/1)

Example
iex> build_function("get_weather", "Get weather for a location", %{
...>   location: {:string, description: "City name"},
...>   units: {:string, enum: ["celsius", "fahrenheit"], description: "Temperature units"}
...> })

Builds a structured output schema from a simple Elixir map or keyword list format.
The input should be a map or keyword list where:
	Keys are field names (atoms)
	Values are either:	A single atom like :string, :number, :boolean, etc.
	A tuple like {:string, description: "...", pattern: "..."}
	For arrays: {:array, :string} or {:array, %{field: :type}}



When using keyword lists, the order of fields is preserved in the required array.
When using maps, fields are sorted alphabetically in the required array.
If the root schema is an array, it will be automatically wrapped in an object
to comply with OpenAI's Structured Outputs requirements.

  


        

      


  

    
OpenAI.Responses.Stream 
    



      
Streaming functionality for the Responses library.
This module provides functions for streaming responses from the OpenAI API,
allowing you to process data as it arrives rather than waiting for the complete response.
Stream Processing Helpers
For processing streamed data, use these helpers:
	text_deltas/1 - Extracts text content from event streams
	json_events/1 - Extracts JSON parsing events from response streams

Examples
# Extract text from streaming response
text = Responses.stream(input: "Write a story")
       |> Responses.Stream.text_deltas()
       |> Enum.join()

# Process JSON events directly from response stream
Responses.stream(input: "Generate data", schema: %{name: :string})
|> Responses.Stream.json_events()
|> Stream.each(&IO.inspect/1)
|> Stream.run()

      


      
        Summary


  
    Functions
  


    
      
        delta(f)

      


        Calls the provided function on all the text chunks received from the server.



    


    
      
        json_events(stream)

      


        Extracts JSON parsing events from a response stream.



    


    
      
        stream(options)

      


        Returns a Stream that yields chunks from the OpenAI API.



    


    
      
        stream_with_callback(callback, options)

      


        Stream a response from the OpenAI API with a callback function.



    


    
      
        text_deltas(stream)

      


        Extracts text deltas from a stream, ignoring errors and other event types.

Calls the provided function on all the text chunks received from the server.
This helper handles the wrapped results and only processes successful text delta events.
Errors and other events are silently ignored.
Examples
Responses.create(
  input: "Write a story",
  stream: Responses.Stream.delta(&IO.write/1)
)

Extracts JSON parsing events from a response stream.
This helper automatically extracts text deltas from the event stream and
converts them into JSON parsing events using incremental parsing.
Events
Returns a Stream that yields JSON parsing events:
	:start_object - Beginning of a JSON object
	:end_object - End of a JSON object
	:start_array - Beginning of a JSON array
	:end_array - End of a JSON array
	{:string, value} - A string value
	{:integer, value} - An integer value
	{:float, value} - A float value
	{:boolean, value} - A boolean value
	:null - A null value
	:colon - Colon between key and value
	:comma - Comma between elements

Examples
# Process JSON events directly from API response
events = Responses.stream(
  input: "Give me 10 U.S. presidents",
  schema: %{presidents: {:array, %{name: :string, birth_year: :integer}}}
)
|> Responses.Stream.json_events()
|> Enum.to_list()

# Process events as they arrive
Responses.stream(
  input: "Generate product catalog",
  schema: %{products: {:array, %{name: :string, price: :number}}}
)
|> Responses.Stream.json_events()
|> Stream.each(fn event ->
  IO.inspect(event, label: "JSON Event")
end)
|> Stream.run()

  



  
    
      
    
    
      stream(options)



        
          
        

    

  


  

Returns a Stream that yields chunks from the OpenAI API.
This function returns an Enumerable that yields results wrapped in tuples:
{:ok, chunk} for successful chunks or {:error, reason} for parsing errors.
Examples
# Process all results with error handling
Responses.stream(input: "Hello")
|> Enum.each(fn
  {:ok, chunk} -> IO.inspect(chunk)
  {:error, reason} -> IO.puts("Error: #{inspect(reason)}")
end)

# Get only text deltas, ignoring errors
text = Responses.stream(input: "Write a story")
       |> Stream.filter(fn
         {:ok, %{event: "response.output_text.delta"}} -> true
         _ -> false
       end)
       |> Stream.map(fn {:ok, chunk} -> chunk.data["delta"] end)
       |> Enum.join()

  



  
    
      
    
    
      stream_with_callback(callback, options)



        
          
        

    

  


  

Stream a response from the OpenAI API with a callback function.
Takes a callback function that will be called for each parsed chunk of the stream.
The callback receives results wrapped in tuples:
{:ok, %{event: "event_type", data: %{...}}}
# or
{:error, reason}
The callback should return :ok to continue streaming or {:error, reason} to stop.
Options
Accepts the same options as Responses.create/1, and automatically adds stream: true parameter.
Examples
# Simple debugging - print all results
Responses.Stream.stream_with_callback(&IO.inspect/1, input: "Hello")

# Process only text deltas using the delta/1 helper
Responses.Stream.stream_with_callback(
  Responses.Stream.delta(&IO.write/1),
  input: "Write a story"
)

# Custom processing with error handling
Responses.Stream.stream_with_callback(fn
  {:ok, %{event: "response.output_text.delta", data: %{"delta" => text}}} ->
    IO.write(text)
    :ok
  {:ok, %{event: "response.completed"}} ->
    IO.puts("
Stream complete!")
    :ok
  {:error, reason} ->
    IO.puts("
Stream error: #{inspect(reason)}")
    :ok  # Continue despite error
  _ ->
    :ok
end, input: "Tell me a joke")

Extracts text deltas from a stream, ignoring errors and other event types.
This helper transforms a raw event stream into a text-only stream by filtering
for response.output_text.delta events and extracting their delta content.
All errors and non-text events are silently ignored.
Examples
# Get text stream from API response
text = Responses.stream(input: "Write a story")
       |> Responses.Stream.text_deltas()
       |> Enum.join()

# Process text incrementally
Responses.stream(input: "Count to 10")
|> Responses.Stream.text_deltas()
|> Stream.each(&IO.write/1)
|> Stream.run()

  


        

      


  

    
OpenAI.Responses.Error exception
    



      
Exception module for OpenAI API errors.
This module represents errors that can occur when making requests to the OpenAI API.
It includes standard OpenAI error fields as well as HTTP status codes.
Fields
	:message - Human-readable error description
	:code - OpenAI error code (e.g., "invalid_api_key")
	:param - The parameter that caused the error (if applicable)
	:type - OpenAI error type (e.g., "invalid_request_error")
	:status - HTTP status code from the response

Examples
iex> error = %OpenAI.Responses.Error{
...>   message: "Rate limit exceeded",
...>   code: "rate_limit_exceeded",
...>   type: "rate_limit_exceeded",
...>   status: 429
...> }
iex> OpenAI.Responses.Error.retryable?(error)
true

        retryable?(arg1)

      


        Determines if an error is retryable.

Determines if an error is retryable.
Returns true if the error is due to temporary server issues that may resolve
with retry attempts, such as rate limits, internal server errors, or network
timeouts. Returns false for client errors like malformed requests that will
not succeed on retry.
Retryable conditions
HTTP status codes
	429 - Rate limit or quota exceeded
	500 - Internal server error
	503 - Service overloaded / Slow down

Transport errors
	:timeout - Request timeout
	:closed - Connection closed

Examples
iex> error = %OpenAI.Responses.Error{status: 429}
iex> OpenAI.Responses.Error.retryable?(error)
true

iex> error = %OpenAI.Responses.Error{status: 400}
iex> OpenAI.Responses.Error.retryable?(error)
false

iex> error = %Req.TransportError{reason: :timeout}
iex> OpenAI.Responses.Error.retryable?(error)
true

